

LOGO . MAT

Uma aplicação do processamento de listas

Uma das aplicações do Logo pode consistir na implementação de modelos simples relativos a fenômenos reais. Nesta aplicação será sempre aconselhável começar por um modelo bastante simples de implementar e progressivamente refinar o modelo através da incorporação de mais elementos, e portanto mais parâmetros, no conjunto de procedimentos que o concretiza.

Existe um processo vulgarmente utilizado em Logo quando se pretende aplicar um mesmo conjunto de operações a cada um dos elementos de uma lista. Este processo geral consiste em tomar a lista, retirar o primeiro elemento, aplicar-lhe as operações e continuar o processo com os restantes elementos da lista até esta se esgotar.

Repare-se no seguinte exemplo: pretende-se escrever no ecrã cada uma das letras da lista $K=[A B C]$. A instrução **PRINT** destina-se a escrever no ecrã o input correspondente. Este input é o primeiro elemento da lista K (isto é, **FIRST :K**). E o processo deverá continuar com a nova lista que se obtém retirando esse elemento à anterior (isto é, a lista **BF :K**). Este processo repetir-se-á até que a lista em processamento seja totalmente esvaziada. O procedimento **ESCREVER** resolve a questão:

```
TO ESCREVER :K
IF EMPTY? :K [STOP]
PRINT FIRST :K
ESCREVER BF :K
END
```

A execução do procedimento **ESCREVER** tem a seguinte estrutura:

```
ESCREVER EMPTY? FIRST:K PRINT BF:K
```

1.º [A B C]	não	A	A	[B C]
2.º [B C]	não	B	B	[C]
3.º [C]	não	C	C	[]
4.º []	sim	—	—	—

Uma aplicação interessante deste processo de aplicação de um conjunto de operações de certo tipo a todos os elementos de uma lista é a concretização de um modelo simples do processo de desenvolvimento de um foco de incêndio (apresentado nesta mesma revista no artigo sobre Modelos e Aplicações).

O fogo inicia-se no canto inferior esquerdo de um reticulado e desenvolve-se para cima e para a direita de acordo com o modelo.

```
TO FOGO :M
CT CG HT
MAKE "L [[0 0]]
MAKE "P :M
EXECUTAR
END
```

```
TO EXECUTAR
PINTAR :L
MAKE "LI :L
MAKE "L []
IF EMPTY? :LI [PRINT [FOGO EXTINTO!] STOPALL]
PEGAR :LI
EXECUTAR
END
```

```
TO PEGAR :LL
IF EMPTY? :LL [STOP]
MAKE "PO FIRST :LL
IF PRO [MAKE "NOVO LIST FIRST :PO 1+LAST :PO ACTUALIZAR]
IF PRO [MAKE "NOVO LIST 1+FIRST :PO LAST :PO ACTUALIZAR]
PEGAR BF :LL
END
```

```
TO ACTUALIZAR
IF AND NOT MEMBER? :NOVO :LI NOT MEMBER? :NOVO :L [MAKE
"L LPUT :NOVO :L]
END
```

```
TO PINTAR :K
IF EMPTY? :K [STOP]
PU
SETPOS FIRST :K
PD
FD 0
PU
PINTAR BF :K
END
```

```
TO PRO
IF :P > RANDOM 100 [OUTPUT "TRUE]
OUTPUT "FALSE
END
```

continua na pág. 38

